Rolf-Dieter Klein

MuMATH/MuSIMP: ein System für symbolische Arithmetik

Mit MuMATH geht ein Wunschtraum vieler in Erfüllung: Es ist jetzt möglich, symbolisch zu rechnen. In Programmiersprachen wie Basic und Pascal oder mit Taschenrechnern kann nur numerisch gerechnet werden. Integrale, Differentialgleichungen o. ä. lassen sich nur näherungsweise lösen. Mit MuMATH kann nun auch eine Formel als Ergebnis erlangt werden. MuMATH wird von Microsoft vertrieben.

MuMATH/MuSIMP gibt es für das CP/M-Betriebssystem mit 8080- oder Z80-CPUs, zum Beispiel auch in einer abgemagerten Version für den TRS-80. MuMATH besteht aus mehreren Paketen, die von MuSIMP aus geladen werden. *Bild 1* zeigt die einzelnen Programmteile. Die Programme sind hierarchisch gegliedert. Es genügt, den Teil der Programme zu laden, der für eine spezielle Aufgabe benötigt wird. Der Speicherbedarf der einzelnen Pakete ist jeweils angegeben. In einem 60-KByte-System ist es möglich, alle Module auf einmal zu laden.

Bild 1. Verschiedene Teilprogramme von MuMATH



Die Features von MuMATH

MuMATH arbeitet, wie schon gesagt, mit symbolischer Arithmetik. Zahlen, die verwendet werden, können bis zu 600 Stellen lang sein. *Bild 2* zeigt einige Beispiele für das Grundpaket ARITH.-MUS. Mit MuMATH kann auch in verschiedenen Zahlenbereichen gerechnet werden, wie *Bild 3* zeigt. Um symbolisch rechnen zu können, wird das Paket ALGEBRA.ARI geladen. Da nicht immer eindeutig ist, wie gerechnet oder umgeformt werden soll, muß MuMATH durch eine geeignete Einstellung von Variablen mitgeteilt bekommen, wie verfahren werden soll. *Bild 4* und *Bild 5* zeigen, welche Unterscheidungen möglich sind. Die Belegung der Variablen kann jederzeit abgerufen werden (*Tabelle 1*). *Tabelle 2* zeigt nun ein paar einfache Umformungen, die mit MuMATH durchgeführt werden können. MuMATH kann auch trigonometrisch (*Bild 6*) und logarithmisch (*Bild 7*) rechnen. Dabei ist es wichtig zu wissen, daß MuMATH immer exakt rechnet und nur rationale Zahlen kennt, also zum Beispiel SIN(10) nicht weiter auswerten kann, da dieser Ausdruck ein nicht rationales Ergebnis liefert. Mit MuMATH können auch einfache Gleichungssysteme gelöst werden (*Bild 8*). Sehr interessant ist die Fähigkeit, mit Matrizen zu operieren. *Bild 9* zeigt einige Beispiele. Die Ergebnisse sind dabei immer mathematisch exakt. MuMATH beherrscht die Differentialrechnung. *Bild 10* zeigt ein paar Beispiele. Zur Grenzwertberechnung gibt es ebenfalls ein Paket in MuMATH, *Bild 11* zeigt Rechnung damit. PINF ist dabei „+ unendlich" und MINF „− unendlich". SIGMA und PROD sind Summen und Produktfunktionen. Ein weiterer Zusatz ermöglicht es, Taylor-Reihen aufzustellen (*Bild 12*). Ein großer Vorteil von MuMATH ist auch die Fähigkeit zur symbolischen Integration. *Bild 13* zeigt einige Möglichkeiten. Bestimmte Integrale können sogar ausgewertet werden (*Bild 14*). MZERO ist dabei der Wert 0, der durch einen Grenzübergang entstanden ist. Es können auch bestimmte Inte-

Tabelle 1: Die Schaltervariable „Faktor“ stellt Wertebereiche ein

Faktor	Art	z. B.
2	Numerische Werte	4, $\frac{100}{3}$
3	Andere Nicht-Summen	X, EXP(3)
5	Summen	A+B, SIN(X)+Y

grale mit ∞ als Grenze berechnet werden. *Bild 15* zeigt, wie auch eigene Programme entwickelt werden können. Diese Programme werden in MuSIMP erstellt und können die Funktionen von MuMATH mit verwenden. *Bild 16* zeigt, wie das erstellte Programm eine Kurvendiskussion durchführen kann.

Tabelle 2: Die Steuervariablen geben Umformungsrichtungen an

Steuer Variable	Vorbelegt	Wert positiv	Wert negativ
NUMNUM	6	$A*(B+C) \rightarrow A*B+A*C$	$A*B+A*C \rightarrow A*(B+C)$
DENDEN	6	$\frac{1}{A}*\frac{1}{B+C} \rightarrow \frac{1}{A*B+A*C}$	$\frac{1}{A*B+A*C} \rightarrow \frac{1}{A}*\frac{1}{B+C}$
DENNUM	6	$\frac{B+C}{A} \rightarrow \frac{B}{A}+\frac{C}{A}$	$\frac{B}{A}+\frac{C}{A} \rightarrow \frac{B+C}{A}$
NUMDEN	0	$\frac{A}{B+C} \rightarrow \frac{1}{B/A+C/A}$	$\frac{1}{B/A+C/A} \rightarrow \frac{A}{B+C}$
BASEXP	−30	$A\uparrow(B+C) \rightarrow A\uparrow B*A\uparrow C$	$A\uparrow B*A\uparrow C \rightarrow A\uparrow(B+C)$
EXPBAS	30	$(A*B)\uparrow C \rightarrow A\uparrow C*B\uparrow C$	$A\uparrow C*B\uparrow C \rightarrow (A*B)\uparrow C$
PWREXPD	0	$(A+B)\uparrow N \rightarrow A\uparrow N+\ldots+B\uparrow N$	$(A+B)\uparrow -N \rightarrow \frac{1}{A\uparrow N+\ldots B\uparrow N}$

Bild 2. ARITH.MUS-Beispiele

```
2/3+5/8;
@: 31 / 24

?
2↑45;
@: 35184372088832

?
2↑64;
@: 18446744073709551616

?
251!;
@:
8114469214881860408125244525581164862197057731361189473537335492053187606363329
1358169329675812807806218305534565054766460946385991505669652054860004689176234
0808337287505448037728337751512715064266648553413711482053864096073935168807868
7793892311994159870905473712136144536785797191004072048829549630785334171335681
6593931096049156059543537367319591093041170193874715434988241628704272109368861
7485205308997682124247612021055212748800000000000000000000000000000000000000000
000000000000000000000
```

Bild 3. Mögliche Zahlensysteme, die MuMATH verarbeitet

```
RADIX(2);
@: 1010

?
10010101*1010010101;
@: 11000000010111001

?
RADIX(1010);
@: 2

?
RADIX(36);
@: 0A

?
0ALPHA*0BETHA;
@: 3CZWRHVAIS

?
M:0ALPHA*0BETHA;
@: 3CZWRHVAIS

?
M:M↑2;
@: 0BAOE8UETRBOX2XUXCLS

?
M:M/(0ALPHA↑2);
@: 3M8450UEIS

?
M;
@: 3M8450UEIS

?
M↑(1/2);
@: 0BETHA

?
RADIX(0A);
@: 36
```

Bild 4. Belegung der Schaltervariablen

```
FLAGS();
@:
TRGSQ = 0            PWREXPD = 0        BASEXP = -30      EXPBAS = 30
NUMDEN = 0           DENDEN = 6         DENNUM = 6        NUMNUM = 30
PBRCH = TRUE         TRGEXPD = 0        LOGEXPD = 0       LOGBAS = #E
ZEROBASE = FALSE     ZEROEXPT = TRUE
```

Bild 5. ALGEBRA.ARI

```
5*X+7*X↑3+4*X+C;
@: C + 9*X + 7*X↑3

?
EXPD((A*X+B*Y)↑4+GE*6+X);
@: X + 6*GE + 4*B*A↑3*X↑3*Y + 6*B↑2*A↑2*X↑2*Y↑2 + 4*B↑3*A*X*Y↑3 + B↑4*Y↑4 + A↑
4*X↑4

?
EXPD((A+B)↑2);
@: 2*B*A + B↑2 + A↑2

?
FCTR(4*X↑3/G+17*B/Z↑2+C/Z↑3);
@: (G*(C+17*B*Z)+4*X↑3*Z↑3) / (G*Z↑3)
```

Bild 6. TRGPOS, TRGNEG

```
TRGEXPD: -6;
@: -6

?
SIN(3*X);
@: 4*COS(X)↑2*SIN(X) - SIN(X)

?
TRGEXPD:15;
@: 15

?
COS(X)↑2;
@: 1/2 + COS(2*X)/2

?
SIN(7*#PI/3);
@: 3↑(1/2) / 2

?
TRGEXPD:7;
@: 7

?
COS(X)+#I*SIN(X);
@: #E ↑ (#I*X)

?
SIN(X);
@: #I/(2*#E↑(#I*X)) - #E↑(#I*X)*#I/2
```

Bild 7. LOG

```
LN(X*5*G↑2);
@: LN (5*X*G↑2)

?
LOGEXPD:-15;
@: -15

?
3+LN(X)*4;
@: 3 + LN(X↑4)

?
LN(A)+LN(B*C)+H*LN(X);
@: LN (B*A*C*X↑H)
```

Bild 8. SOLVE, Lösung von Gleichungen

```
SOLVE(A*X↑2+B*X+C==0,X);

@: (X == -B/(2*A)+(B↑2/(4*A↑2)-C/A)↑(1/2),
    X == -B/(2*A)-(B↑2/(4*A↑2)-C/A)↑(1/2))

?
SOLVE(X↑5+2*X==4,X);
@: (-4+2*X+X↑5 == 0)

?
SOLVE(X↑5+2*X==0,X);
@: (X == 2↑(1/4)*#I↑(7/2),
    X == 2↑(1/4)*#I↑(5/2),
    X == 2↑(1/4)*#I↑(3/2),
    X == 2↑(1/4)*#I↑(1/2),
    X == 0)

?
SOLVE(LOG(4+X)==3,X);
@: (X == -4+#E↑3)
```

Bild 9. MATRIX, Matrizenarithmetik

```
M1:([1,2,3],[13,21,4],[8,3,1]);
@: ([1, 2, 3],
    [13, 21, 4],
    [8, 3, 1])

?
M1↑-1;
@: ([-9/340, -7/340, 11/68],
    [-19/340, 23/340, -7/68],
    [129/340, -13/340, 1/68])

?
M1.M1↑-1;
@: ([1, 0, 0],
    [0, 1, 0],
    [0, 0, 1])

?
MAT1:([1,2],[A,B]);
@: ([1, 2],
    [A, B])

?
MAT1.MAT1;
@: ([1+2*A, 2+2*B],
    [A+B*A, 2*A+B↑2])

?
MAT1↑-1;
@: ([1+2*A/(B-2*A), -2/(B-2*A)],
    [-A/(B-2*A), 1/(B-2*A)])

?
MAT1:([A,B],[C,D]);
@: ([A, B],
    [C, D])

?
MAT1↑-1;
@: ([B*C/(-B*A*C+A↑2*D)+1/A, -B/(-B*C+A*D)],
    [-C/(-B*C+A*D), 1/(D-B*C/A)])
```

Bild 10. DIF, Lösung von Differentialen

```
TRGEXPD:-6$

?
DIF(1/X↑5,X);
@: -5 / X↑6

?
DIF(SIN(X)/(1+3*X↑2),X);
@: -6*X*SIN(X)/(1+3*X↑2)↑2 + COS(X)/(1+3*X↑2)

?
DIF(LN(A+X),X);
@: 1 / (A+X)

?
DIF(ATAN(SIN(X)),X);
@: COS(X) / (1+SIN(X)↑2)

?
DIF(A*X↑2*F(X),X);
@: 2*A*X*F(X) + A*X↑2*DIF(F(X),X)
```

Bild 11. LIM und SIGMA

```
% BILD 11 %

LIM((X↑3+4*X+5)/(2*X↑3+4+X),X);
@: 5 / 4

?
LIM((X↑3+4*X+5)/(2*X↑3+4+X),X,PINF);
@: 1 / 2

?
SIGMA(1/(N!*X),N,1,4);
@: 41 / (24*X)

?
SIGMA(2↑-X,X,0,PINF);
@: Is SIGN (-LN(2))   -1; 0; or 1; ?
-1; Is SIGN (-LN(2))   -1; 0; or 1; ?
-1; 2 + MZERO

?
PROD((A+B*X),X,2,5);
@: 14*B*A↑3 + 71*B↑2*A↑2 + 154*B↑3*A + 120*B↑4 + A↑4
```

Bild 12. TAYLOR

```
A:TAYLOR(#E↑X,X,0,6);
@: 1 + X + X↑2/2 + X↑3/6 + X↑4/24 + X↑5/120 + X↑6/720

?
X:1;
@: 1

?
EVAL(A);
@: 1957 / 720

?
TAYLOR(SIN(X),X,0,3);
@: 5 / 6

?
A:'A$

? X:'X$
```

Bild 13. INT, Integralrechnung

```
INT(SIN(X),X);
@: - COS(X)

?
INT(#E↑(-X↑2),X);
@: #PI↑(1/2)*ERF(X) / 2

?
INT(1/(X+B),X);
@: LN (B+X)

?
INT(SIN(X)*COS(X),X);
@: -COS(X)↑2 / 2

?
INT(X*LN(B*X),X);
@: X↑2*LN(B)/2 + X↑2*LN(X)/2 - X↑2/4
```

Bild 14. Bestimmte Integrale

```
DEFINT(SIN(X),X,0,#PI/2);
@: 1

?
DEFINT(1/X↑3,X,1,2);


@: 3 / 8

?
DEFINT(1/X↑2,X,1,PINF);
@: 1 + MZERO
```

Bild 15. Beispiel für Anwenderfunktion

```
FUNCTION KURDIS(EX1,EX2),
   PRINTLINE(" KURVENDISKUSSION "),
   PRINTLINE(" 1. NULLSTELLEN "),
   PRTMATH(SOLVE(EX1==0,EX2),0,0,TRUE),
   PRINTLINE(" "),
   PRINTLINE(" 2. GRENZWERTE BEI 0, -UNENDL, + UNDENDL "),
   PRTMATH(LIM(EX1,EX2,0),0,0,TRUE),
   PRINTLINE(" "),
   PRTMATH(LIM(EX1,EX2,MINF),0,0,TRUE),
   PRINTLINE(" "),
   PRTMATH(LIM(EX1,EX2,PINF),0,0,TRUE),
   PRINTLINE(" "),
   PRINTLINE(" 3. EXTREMA "),
   PRTMATH(SOLVE(DIF(EX1,EX2)==0,EX2),0,0,TRUE),
   PRINTLINE(" "),
   PRINTLINE(" 4. WENDEPUNKTE "),
   PRTMATH(SOLVE(DIF(DIF(EX1,EX2),EX2)==0,EX2),0,0,TRUE),
   PRINTLINE(" "),
ENDFUN$
```

Bild 16. Kurvendiskussion mit der neuen Funktion

```
KURDIS(X↑3-X,X);
@: KURVENDISKUSSION
 1. NULLSTELLEN
(X == -1,
    X == 1,
    X == 0)
 2. GRENZWERTE BEI 0, -UNENDL, + UNDENDL
0
MINF
PINF
 3. EXTREMA
(X == -1/3↑(1/2),
    X == 1/3↑(1/2))
 4. WENDEPUNKTE
(X == 0)


?
KURDIS(SIN(X)/X,X);
@: KURVENDISKUSSION
 1. NULLSTELLEN
(X == ASIN(0))
 2. GRENZWERTE BEI 0, -UNENDL, + UNDENDL
1
?
?
 3. EXTREMA
(X*COS(X)-SIN(X) == 0)
 4. WENDEPUNKTE
(-2*X*COS(X)-X↑2*SIN(X)+2*SIN(X) == 0)


?

STOP$

?
RDS (" ") $

? ↑L
;

@:
```